

Analize în curent continuu cu PSPICE

Analizele standard care pot fi efectuate în programul PSPICE sunt următoarele:

- .OP (calculul punctului static de funcționare);
- .DC (analiza în curent continuu);
- .TF (calculul câștigului de curent continuu în semnal mic);
- .SENS (analiza sensibilităților în curent continuu);
- .AC (analiza în domeniul frecvență);
- .NOISE (analiza de zgomot);
- .TRAN (analiza în domeniul timp);
- .FOUR (calculul componentelor Fourier).

În această lucrare de laborator sunt prezentate comenzile pentru analizele în curent continuu.

.OP Punctul static de funcționare

Forma generală a comenzii este:

.OP

Comanda .OP determină calcularea și furnizarea unor informații detaliate despre punctul static de funcționare. Astfel, în afară de lista tensiunilor nodale, a curenților prin sursele de tensiune și puterea totală disipată, se prezintă și parametrii de semnal mic (liniarizați) ai tuturor dispozitivelor semiconductoare și surselor neliniare comandate.

Punctul static de funcționare este calculat de către programul PSpice indiferent de prezența comenzii .OP, această informație fiind disponibilă când se precizează alte tipuri de analize ale circuitului. Când se dorește ca programul să calculeze numai punctul static de funcționare, fără efectuarea altor analize, atunci este necesară includerea comenzii .OP. Toate informațiile legate de punctul static de funcționare sunt tipărite în mod text în fișierul de ieșire (*.out).

Exemplu: Calculul PSF al unui circuit simplu cu diodă semiconductoare.

```
*Calculul PSF
V1 1 0 5
R1 1 0 100
R2 1 2 100
D1 2 0 D1N4148
.MODEL D1N4148 D(Is=2.682n N=1.836 Rs=.5664 Ikf=44.17m Xti=3
+Eg=1.11 Cjo=4p M=.03333 Vj=.5 Isr=1.565n Nr=2 Bv=100 Ibv=100u Tt=11.54n)
.OP
.END
```

Rezultatele furnizate de Pspice ca urmare a comenzii .OP sunt tipărite în fișierul de ieșire sub forma următoare:

SMALL SIGNAL BIAS SOLUTION

TEMPERATURE = 27.000 DEG C

NODE	VOLTAGE	NODE	VOLTAGE
(1)	5.0000	(2)	.8206

VOLTAGE	SOURCE	CURRENTS
NAME	CURRENT	
V1	-9.179E-02	

TOTAL	POWER	DISSIPATION	4.59E-01	WATTS
-------	-------	-------------	----------	-------

Probleme care apar la calculul PSF

Reușita calculării punctului static de funcționare este esențială pentru analiza circuitului. Dacă din diverse motive programul nu reușește să calculeze PSF, atunci nu pot fi efectuate nici celelalte analize.

Algoritmul de calcul al PSF presupune plecarea de la un punct de start și calcularea unui șir de soluții care, în anumite condiții, converg spre soluția finală. Dacă acest algoritm nu converge, atunci programul PSPICE încearcă să calculeze PSF prin metoda de incrementare a surselor de alimentare. Dacă nici cu acest algoritm nu reușește să determine soluția de punct static, programul tipărește un mesaj corespunzător în fișierul de ieșire.

Utilizatorul poate veni în sprijinul algoritmului de calcul al PSF introducând estimarea sa inițială cea mai bună, de la care să starteze algoritmul. Acest lucru este posibil prin intermediul comenzii .NODESET. Dacă punctul de start este apropiat de soluția finală, atunci există șanse foarte mari ca algoritmul să convergă.

Utilizarea comenzii .NODESET este recomandată a se folosi întodeauna la simularea repetată a circuitelor electronice complexe la care, în mod obișnuit, calculul PSF consumă un timp considerabil.

Comanda .NODESET

Forma generală a comenzii .NODESET este:

```
.NODESET <V(<nod1>[,<nod2>])=<valoare>>*
.NODESET I(<nume_bobină>)=<valoare>
```

Exemple:

- 1) .NODESET V(5)=2.6V V(12)=-1.7V I(L1)=1uA - inițializează tensiunile V(5) și V(12), precum și curentul prin bobina L1 pentru algoritmul de calcul al punctului static;
- 2) .NODESET V(IN1, IN2)=3mV V(9, 3)=7.25V - inițializează tensiunile dintre nodurile IN1 și IN2, respectiv nodurile 9 și 3 pentru ușurarea calculului punctului static.

Comanda .NODESET ajută la calculul punctului static de funcționare prin furnizarea unor valori de tensiuni nodale și/sau curenți prin bobine, apropiate de soluția reală. Acestea reprezintă "ghicirea" soluției de punct static și constituie punctul de start al algoritmului de calcul al PSF. "Ghicirea" soluției este o problemă delicată și solicită utilizatorului să cunoască bine funcționarea circuitului sau să aplice diferite metode rapide pentru a o determina aproximativ. Folosind această comandă se poate obține o reducere importantă a

timpului de calcul a PSF și chiar evitarea divergenței algoritmului. Ea are efect asupra calculului PSF corespunzător analizelor de semnal mic și tranzitorie, precum și asupra primului pas al analizei de curent continuu.

Diferența dintre comenzile .NODESET și .IC este aceea că la prima, tensiunile reprezintă punctul de start pentru algoritmul de calcul a punctului static, în timp ce la comanda .IC tensiunile precizate se regăsesc în soluția de punct static și constituie o condiție inițială pentru analiza de semnal mic sau tranzitorie. Dacă aceste două comenzi există simultan, atunci comanda .NODESET este ignorată.

.DC Analiza în curent continuu

Formele generale ale comenzii sunt:

`.DC [LIN] <nume_variabilă> <val_start> <val_finală> <val_increm> [înlănțuire]`

`.DC <tip_variație_logaritmică> <nume_variabilă> <val_start> <val_finală> <nr_puncte> +[înlănțuire]`

`.DC <nume_variabilă> LIST <valori>* [înlănțuire]`

Exemple:

- 1) `.DC Vin -5V 5V 10mV` - determină o analiză de curent continuu în care se variază liniar tensiunea sursei Vin de la -5V la +5V, cu un pas de 10mV;
- 2) `.DC LIN Iin 100mA 1A 1mA` - determină o analiza de curent continuu în care se variază liniar curentul sursei Iin de la 100mA la 1A, cu pasul de 1 mA;
- 3) `.DC VCE 0 10V 0.5V IB 50uA 1mA 200uA` - determină o analiză de curent continuu înlănțuită, variindu-se liniar tensiunea sursei VCE de la 0V la 10V, cu pasul de 0.5V, pentru fiecare valoare a curentului sursei IB, variat liniar de la 50uA la 200uA, cu un pas de 50uA;
- 4) `.DC RES RMOD(R) 0.5 1.5 0.01` - determină o analiză de curent continuu în care se variază liniar parametrul de model R al modelului RMOD de rezistor, între 0.5 și 1.5, cu pasul de 0.01;
- 5) `.DC DEC NPN QNPN(IS) 1E-18 1E-15 5` - determină o analiză de curent continuu variindu-se logaritm, între 1E-18A și 1E-15A, cu 5 puncte de calcul pe decadă, parametrul de model IS al modelului QNPN de tranzistor bipolar NPN;
- 6) `.DC TEMP -50 125 5` - determină o analiză de curent continuu variindu-se liniar temperatura de la -50°C la 125°C, cu pasul de 5°C;
- 7) `.DC DEC VCE 1mV 10V 10 IB LIST 100uA 200uA 300uA 400uA` - determină o analiză de curent continuu înlănțuită, variindu-se logaritm tensiunea sursei VCE între 1E-3V și 10V, cu 10 puncte pe decadă, pentru fiecare din valorile curentului sursei IB specificate în lista de valori;
- 8) `.DC PARAM ALFA 0 1 10m` - determină o analiza de curent continuu variindu-se liniar parametrul global ALFA de la 0 la 1, cu pasul de 0.01.

Comanda .DC realizează o analiză de curent continuu a unui circuit. Aceasta constă în determinarea punctului static în funcție de variabila precizată cu <nume_variabilă>, a cărei valoare este cuprinsă între <val_start> și <val_finală>. Exemplele 1, 2, 3, 4, 6 și 8 corespund primei forme generale de descriere, exemplul 5 corespunde formei a doua, iar exemplul 7 corespunde formelor a doua și a treia.

La prima formă (variație liniară), <val_start> poate fi mai mare decât <val_finală>, valoarea incrementării <val_increm> trebuie să fie totdeauna mai mare ca zero, iar cuvântul cheie LIN poate lipsi. La forma a doua (variație logaritmică), tipul acesteia este DEC (pe decade) sau OCT (pe octave). <nr_puncte> reprezintă numărul de puncte de calcul pe decadă sau octavă. La forma a treia se efectuează analiza de curent continuu pentru lista de valori ale variabilei <nume_variabilă>, precizată după cuvântul cheie LIST. După finalizarea analizei .DC variabilei îi este atribuită valoarea avută înaintea variației.

Dacă se specifică [înlănțuire] (exemplele 3 și 7), analiza de curent continuu este multiplă, adică aceasta se repetă pentru fiecare valoare a variabilei din [înlănțuire]. Se pot obține astfel familii de caracteristici de curent continuu.

Observație: Dacă se utilizează analiza de curent continuu cu înlănțuire, pentru a putea reprezenta grafic rezultatele este obligatoriu să se includă în fișierul circuitului comanda .PROBE.

<nume_variabilă> poate fi de tipul:

- Sursă independentă - nume de sursă independentă de tensiune sau curent (ex. 1, 2, 3);
- Parametru de model - un tip de model și nume de model, urmat între paranteze de numele parametrului de model (exemplele 4 și 5). Nu pot fi folosiți parametrii L și W de la MOSFET, precum și nici un parametru de temperatură, cum ar fi TC1 și TC2 de la rezistoare;
- Temperatură - se specifică cu ajutorul cuvântului TEMP;
- Parametru global - se scrie cuvântul cheie PARAM, urmat de numele parametrului global. Parametrul global trebuie inițializat în fișierul circuitului cu ajutorul comenzii .PARAM.

În fișierul circuitului, valorile diverselor componente ale circuitului pot fi specificate prin intermediul unui parametru global sau al unei expresii în care intervine un parametru global. În acest caz, parametrul global sau expresia este încadrată de acolade {...}.

Rezultatele analizei .DC pot fi reprezentate grafic prin intermediul comenzii .PROBE, sau pot fi tipărite și eventual reprezentate grafic în fișierul de ieșire prin intermediul comenzilor .PRINT, respectiv .PLOT.

Exemplu:

R1 2 6 {RVAL}; rezistența rezistorului R1 are valoarea furnizată de parametrul global RVAL


.PARAM RVAL=1K

*comanda de mai sus definește RVAL ca parametru global și îl inițializează cu valoarea 1000

.DC PARAM RVAL 1K 2K 10

Aplicație rezolvată:

Pentru circuitul divizor de tensiune din figura alăturată, vom determina dependența în curent continuu a tensiunii de ieșire în funcție de valorile rezistorului R2, în intervalul [1, 9] Kohmi.



Fișierul circuitului este următorul:

*Analiza DC

Vin 1 0 5


R1 1 2 5k

R2 2 0 {R2VAL}

.PARAM R2VAL=5k

```
.DC PARAM R2VAL 1k 9k 80
.PROBE
.END
```

Dependența tensiunii de ieșire V(2) în funcție de R2 este prezentată în figura de mai jos:



În analiza de curent continuu de mai sus, varierea valorii rezistorului R2 putea fi realizată și prin intermediul parametrului de model R (multiplicator de rezistență). În acest caz, pentru rezistorul R2 trebuie definit un model. Dacă lui R2 i se atribuie valoarea 5k, pentru a varia valoarea rezistorului între 1k și 9k, trebuie să variem parametrul de model R între 0.2 și 1.8, cu pasul de 0.016 (pentru 101 puncte de calcul). Modificările efectuate sunt următoarele:

```
R2 2 0 RMOD 5K
.MODEL RMOD RES(R=1)
.DC RES RMOD(R) 0.2 1.8 0.016
```

Mai există și posibilitatea ca parametrului de model R să i se atribuie valoarea în comanda .MODEL prin intermediul unui parametru global. Rezultă că variația rezistorului R2 se va face prin intermediul parametrului de model R care, la rândul lui, are valoarea specificată de un parametru global. Această posibilitate este descrisă de următoarele linii:

```
R2 2 0 RMOD 5K
.MODEL RMOD RES(R={RP})
.PARAM RP=1
.DC PARAM RP 0.2 1.8 0.016
```

Dezavantajul ultimelor două posibilități este acela că la reprezentarea grafică a caracteristicii de curent continuu, prin intermediul comenzii .PROBE, pe axa X apare implicit intervalul de variație al parametrului de model R și nu intervalul în care a fost variată valoarea rezistorului R2.

.TF Transferul în curent continuu de semnal mic

Forma generală a comenzii este:

```
.TF <variabilă_ieșire> <nume_sursă_intrare>
```

Exemple:

1) `.TF V(7,3) VIN` - calculează și afișează câștigul în curent continuu, de semnal mic, $V(7,3)/VIN$, precum și rezistențele de intrare (în nodul în care este conectată sursa VIN) și de ieșire (între nodurile 7 și 3) ale circuitului;

2) `.TF I(VNUL) Iin` - calculează și afișează câștigul în curent continuu, de semnal mic, $I(VNUL)/Iin$, precum și rezistențele de intrare (la Iin) și de ieșire (la VNUL).

Comanda `.TF` determină calculul funcției de transfer de semnal mic în curent continuu prin liniarizarea circuitului în jurul punctului static. În fișierul de ieșire sunt tipărite câștigul de la sursa de intrare <nume_sursă_intrare> la variabila de ieșire, <variabilă_ieșire>, precum și rezistențele de intrare și ieșire. Rezultatele comenzii `.TF` nu pot fi reprezentate grafic cu ajutorul comenzii `.PROBE`.

Dacă variabila de ieșire este un curent, aceasta trebuie să fie curentul printr-o sursă independentă de tensiune (eventual de valoare nulă).

.SENS – Calculul sensibilităților

Forma generală a comenzii este:

`.SENS <variabilă_ieșire>*`

Exemplu:

`.SENS V(3) V(5,2) I(VXX)`

Comanda `.SENS` are ca efect determinarea sensibilităților în curent continuu, de semnal mic prin liniarizarea circuitului în jurul punctului static. Pentru fiecare din variabilele de ieșire specificate în comanda `.SENS` se calculează sensibilitățile în raport cu toate valorile și parametrii de model ai dispozitivelor din circuit. Valorile acestor sensibilități sunt tipărite în fișierul de ieșire.

Programul PSPICE calculează sensibilități pentru următoarele tipuri de dispozitive: rezistoare, surse independente, comutatoare, diode și tranzistoare bipolare.

Rezultatele comenzii `.SENS` nu pot fi reprezentate grafic cu ajutorul comenzii `.PROBE`.

Aplicatia 1:

1) Editați în PSPICE circuitul din Figura 1(a), ce constă dintr-o sursă de curent continuu $I_1=30\text{mA}$ și un rezistor neliniar RN. Rezistorul neliniar RN are caracteristica I-V reprezentată în Figura 1(b) și se va modela cu o sursă de curent comandată de tensiunea de la propriile terminale. Explicați necesitatea prezenței rezistorului $R_1=100\text{M}\Omega$ în vederea simulării circuitului.




Figura 1

2) Circuitul din Figura 1(a) are 3 puncte statice de funcționare aflate la intersecțiile dintre caracteristica I-V a rezistorului neliniar și dreapta orizontală $I_{dc}=30\text{mA}$ (punctele A, B și C). Determinați prin simulare punctul de funcționare al circuitului (tensiunea $V(1)$). Punctul de funcționare este tipărit în fișierul de ieșire. Care din cele trei puncte a fost determinat de către simulator?

3) Reluați determinarea punctului de funcționare al circuitului impunând simulatorului prin intermediul comenzii .NODESET diverse valori ale tensiunii $V(1)$ ca punct de start pentru algoritmul de calcul al PSF. Pentru fiecare punct de start specificat în comanda .NODESET notați în tabelul următor soluția determinată de către simulator:

.NODESET V(1)=	0.5	2	2.7	3.5	4.8	6
PSF V(1) determinat de simulator						
Punctul din Figura 1(b) corespunzător soluției						


Ce concluzii se pot extrage în urma acestor simulări? Există o legătură între punctul de start și soluția determinată de către simulator?

Aplicatia 2:

1) Să se descrie în PSPICE circuitul din figura următoare. Modelul BC107A al tranzistoarelor Q1 și Q2 se apelează din fișierele bibliotecă de modele ale programului. Descrierea PSPICE a acestor două tranzistoare este următoarea:

```
Q1 2 1 4 BC107A
Q2 3 0 4 BC107A
.LIB NOM.LIB
```

Modelul BC107A este definit în fișierul bibliotecă sub forma următoare:
 .MODEL BC107A NPN(*parametri_model*).



2) Pentru circuitul din figură să se efectueze următoarele:

a) Să se determine punctul static de funcționare. Se vor nota valorile tensiunilor din nodurile circuitului, valorile curenților $I_C(Q1)$ și $I_C(Q2)$, precum și puterea absorbită din sursele de alimentare.

b) O analiză de curent continuu în care V_{in} se variază în intervalul $(-1V, +1V)$. Pasul de incrementare se va considera astfel încât numărul total de puncte de calcul să fie 101. În urma analizei vizualizați:

i. dependența tensiunii de ieșire în funcție de tensiunea de intrare (caracteristica de transfer în c.c.);

ii. curenții de colector ai tranzistoarelor.

c) O analiză de curent continuu în care să se varieze valoarea rezistenței $R3$ în intervalul $(1K, 5K)$. Vizualizați dependența tensiunii de ieșire și a tensiunii din nodul 4 în funcție de $R3$.

d) O analiză de curent continuu în care se variază parametrul de model BF al tranzistoarelor NPN cu numele de model BC107A, în intervalul (50 , 150).

e) O analiză de curent continuu în care se variază temperatura de la -25°C la +75°C. Cum variază tensiunea de ieșire cu temperatura?

f) Să se efectueze analiza pentru calculul transferului de c.c. în semnal mic (.TF) și să se identifice rezultatele.

g) Să se efectueze analiza pentru calculul sensibilității de c.c. în semnal mic (.SENS) a tensiunii de ieșire și să se identifice rezultatele.